NUSC Technical Memorandum 831172
30 December 1983

A Subroutine Package for the Efficient Solution of the Eigenproblem of Real Symmetric Toeplitz Matrices

Marvin J. Goldstein
William A. Babson
Computer and Information Services Department

REFERENCE ONLY

Naval Underwater Systems Center
Newport, Rhode Island/New London, Connecticut

Report Documentation Page

Form Approved
OMB No. 0704-0188

1. REPORT DATE	2. REPORT TYPE	3. DATES COVERED
30 DEC 1983	**Technical Memo**	**30-12-1983 to 30-12-1983**

4. TITLE AND SUBTITLE
A Subroutine Package for the Efficient Solution of the Eigenproblem of Real Symmetric Toeplitz Matrices

5a. CONTRACT NUMBER

5b. GRANT NUMBER

5c. PROGRAM ELEMENT NUMBER

6. AUTHOR(S)
Marvin Goldstein; William Babson

5d. PROJECT NUMBER
771Y00

5e. TASK NUMBER

5f. WORK UNIT NUMBER

7. PERFORMING ORGANIZATION NAME(S) AND ADDRESS(ES)
Naval Underwater Systems Center,New London,CT,06320

8. PERFORMING ORGANIZATION REPORT NUMBER
TM 831172

9. SPONSORING/MONITORING AGENCY NAME(S) AND ADDRESS(ES)

10. SPONSOR/MONITOR'S ACRONYM(S)

11. SPONSOR/MONITOR'S REPORT NUMBER(S)

12. DISTRIBUTION/AVAILABILITY STATEMENT
Approved for public release; distribution unlimited

13. SUPPLEMENTARY NOTES
NUWC2015

14. ABSTRACT
We discuss a subroutine package for solving the eigenproblem of bisymmetric Toeplitz matrices. When the eigenvalues and eigenvectors of a bisymmetric Toeplitz matrix are computed, storage requirements and execution time are reduced by taking advantage of the special structure of the matrix.

15. SUBJECT TERMS
bisymmtric Toeplitz matrices

16. SECURITY CLASSIFICATION OF: a. REPORT	b. ABSTRACT	c. THIS PAGE	17. LIMITATION OF ABSTRACT	18. NUMBER OF PAGES	19a. NAME OF RESPONSIBLE PERSON
unclassified	**unclassified**	**unclassified**	**Same as Report (SAR)**	**16**	

TM 831172

NAVAL UNDERWATER SYSTEMS CENTER
NEW LONDON LABORATORY
NEW LONDON, CONNECTICUT 06320

Technical Memorandum

A SUBROUTINE PACKAGE FOR THE EFFICIENT SOLUTION OF THE EIGENPROBLEM OF REAL SYMMETRIC TOEPLITZ MATRICES

Date: December 30, 1983

Prepared by: Marvin J. Goldstein

William A. Babson

Approved for public release; distribution unlimited

ABSTRACT

We discuss a subroutine package for solving the eigenproblem of bisymmetric Toeplitz matrices. When the eigenvalues and eigenvectors of a bisymmetric Toeplitz matrix are computed storage requirements and execution time are reduced by taking advantage of the special structure of the matrix.

ADMINISTRATIVE INFORMATION

This memorandum was prepared under Job Order No. 771Y00, Special Projects and Studies. The authors are located at the Naval Underwater Systems Center, New London, Connecticut, 06320.

ACKNOWLEDGEMENT

The authors are grateful to Dr. John Ianniello (Code 3212) for technically reviewing this memorandum.

INTRODUCTION

An n-th order real symmetric matrix $M=(M_{jk})$ for which

$$M_{jk} = M_{j+1,\ k+1} \tag{1}$$

is called a bisymmetric Toeplitz matrix. Bisymmetric Toeplitz matrices arise, for example, as correlation matrices in the study of discrete stationary random processes. Each element of the correlation matrix is a function of j-k; that is,

$$M_{jk} = M_{j-k} = M_{k-j}$$

(e.g. $M_{jk} = \cos(j-k)c$, where c is a real number.)

In [1], using the bisymmetry of matrix M, Goldstein shows that the eigenvalues and eigenvectors of matrix M can be computed from two real symmetric matrices one-half its size, if it is of even order. In this memorandum, we provide a FORTRAN subroutine package that is an implementaiton of this reduction. By using this package, one reduces both computer time and storage ordinarily needed to calculate the eigenvalues and eigenvectors of bisymmetric Toeplitz matrices.

ALGORITHM

The sub-program package in the Appendix computes the eigenvalues and (optionally) the eigenvectors of an even order, n, real symmetric matrix M, whose coefficients satisfy Eq. (1), from two real symmetric matrices:

$$\tilde{M}_{jk}(+) = M_{jk} + M_{j,n+1-k} = \tilde{M}_{kj}(+) \qquad (2)$$

$$(k=1, \ldots, n/2;\ j=1, \ldots, k)$$

$$\tilde{M}_{jk}(-) = M_{jk} - M_{j,n+1-k} = \tilde{M}_{kj}(-) \qquad (3)$$

one-half the size of matrix M, by [1, 2].

The matrices $\tilde{M}(+)$ and $\tilde{M}(-)$ are computed from the first row elements of matrix M. For example, by Property (1), Formula (2) reduces to

$$\tilde{M}_{jk}(+) = M_{1,k-(j-1)} + M_{1,n-(k+j)+2} \qquad (4)$$

$$= \tilde{M}_{kj}(+) \qquad k=1, \ldots, n/2;\ j=1, \ldots, k$$

where addition of the coefficients of the first row of matrix M is replaced by subtraction when computing $\tilde{M}(-)$.

Forming M~(+) first, the subroutine package calculates the eigenvalues and (optionally) the eigenvectors of M~(+). Then M~(-) is formed and its eigenvalues and (optionally) eigenvectors are computed.

The eigenvalues and eigenvectors of matrices M~(+) and M~(-) determine the eigenvalues and eigenvectors of matrix M. If z is an eigenvector of M~(+) with corresponding eigenvalue d, and J is the identity matrix of order n/2 with its columns written in reverse order, then

$$M \begin{bmatrix} z \\ Jz \end{bmatrix} = d \begin{bmatrix} z \\ Jz \end{bmatrix}. \tag{5}$$

Similarly, if z is an eigenvector of M~(-) with corresponding eigenvalue d, then

$$M \begin{bmatrix} z \\ -Jz \end{bmatrix} = d \begin{bmatrix} z \\ -Jz \end{bmatrix}. \tag{6}$$

STORAGE CONSIDERATIONS

Note that in (5) and (6), since premultiplying a vector by matrix J just reverses the order of the vector's components, all the information about the eigenvectors of matrix M is given by the eigenvectors of matrices M~(+) and M~(-); therefore, only n/2 of the components of each eigenvector of matrix M have to be stored, and a fortiori, only n**2/2 entries (instead of n**2) are required to store all the eigenvectors of matrix M.

Furthermore, since the eigenvalues and eigenvectors of M~(+) are computed independently of those for M~(-), matrices M~(+) and M~(-) may share the same storage area for a matrix of order n/2. Therefore, the reduction requires a maximum of (3/4)n**2 + n storage locations to accomodate the matrices M~(+), M~(-) and their eigenvalues and eigenvectors, instead of the 2n**2 + n required when the eigensystem is solved directly from matrix M. In particular, only (1/4)n**2 + n storage locations are used to accomodate M~(+) and M~(-) and their eigenvalues, if only the eigenvalues of matrix M are required.

TIMING

Computational results indicate that the eigenproblem for a large matrix M can be solved by computer at least four times faster this way than directly from matrix M. For example, the following table gives representative execution times (in seconds) to compute the eigenvalues of bisymmetric Toeplitz matrices of orders 16, 32, 64, 128, 256 by the reduction SUBROUTINE

MATRED (in the Appendix) and directly by the IMSL library SUBROUTINE EIGRS [3] on the VAX 11/780 in double precision arithmetic.

Order (n)	MATRED (using EIGRS)	EIGRS
16	.03	.07
32	.13	.38
64	.73	2.64
128	4.97	21.00
256	38.53	248.63

The following table gives representative execution times (in seconds) for a single precision version of SUBROUTINE MATRED and a single precision version of EIGRS on the UNIVAC 1100/62:*

Order (n)	Single Precision MATRED	Single Precision EIGRS
16	.02	.05
32	.10	.33
64	.60	2.24
128	4.08	16.76
256	30.16	129.04

* Since only a single precision version of EIGRS is available in the IMSL library on the UNIVAC, the UNIVAC version of MATRED is single precision.

In multi-program computer environments, variations in the execution time of a program can occur due to differences in machine load. Therefore, in order to smooth out these variations, subroutines MATRED and EIGRS were run a number of times on different matrices of order n and the average execution time for each order was recorded in the tables.

USER PROCEDURE

The calling sequence of SUBROUTINE MATRED (double precision version) is defined in its commentary in the Appendix. The relocatable code, MATRD.OBJ, is in the VAX 11/780 directory [MJG] on node 2. The UNIVAC single precision version of SUBROUTINE MATRED is in the UNIVAC 1100/62 (node 2) file MGOLDSTN*MATRD and has the element name MATRD. It's calling sequence is the same as the double precision version on the VAX, but none of the arguments in the sequence are double precision.

SUMMARY

A subroutine package for computing the eigenvalues and (optionally) the eigenvectors of a bisymmetric Toeplitz matrix is available which reduces storage requirements and execution time by taking advantage of the special structure of the matrix. In particular, if only the eigenvalues of the matrix are required storage requirements for arrays and execution time are reduced by 75% when the matrix is sufficiently large.

REFERENCES

1. Goldstein Marvin, "Reduction of the Eigenproblem for Hermitian Persymmetric Matrices," Math, Comp., v. 38, Jan. 1974

2. Goldstein Marvin, "Further Decomposition of the Pseudoinverse and Eigensystem of a Hermitian Persymmetric Matrix," ACM SIGNUM Newsletter, v. 13, no. 1, March 1978.

3. "EIGRS," IMSL Library for VAX-11/780, Eighth Edition, 1980.

APPENDIX

```
      SUBROUTINE MATRED(IVEC,MR1,N,IFULST,MTILFL,IFUL,MTILSY,D,Z,IZ,WK)
C     ******************************************************************
C            A SUBROUTINE PACKAGE FOR THE EFFICIENT SOLUTION
C      OF THE EIGENPROBLEM OF REAL SYMMETRIC TOEPLITZ MATRICES
C
C                                  BY
C
C                  MARVIN GOLDSTEIN AND WILLIAM BABSON
C
C     ***      THIS SUB-PROGRAM PACKAGE COMPUTES THE EIGENVALUES AND     ***
C     ***(OPTIONALLY) THE EIGENVECTORS OF AN EVEN ORDER (n) REAL SYM-    ***
C     ***METRIC TOEPLITZ MATRIX:                                         ***
C
C                M    = M                                       (1)
C                 jk     j+1,k+1
C
C                     = M
C                        kj
C
C     ***FROM TWO REAL SYMMETRIC MATRICES:                               ***
C
C                M~(+) =   M     + M          = M~ (+)          (2)
C                 jk        jk      j,n+1-k      kj
C
C                              (k=1, ...., n/2; j=1, ...., k)
C
                 M~ (-) = M    - M            = M~ (-)          (3)
C                 jk        jk      j,n+1-k      kj
C
C     ***ONE-HALF THE SIZE OF MATRIX M, BY [1, 2].                       ***
C
C     ***      THE MATRICES M~(+) AND M~(-) CAN BE COMPUTED FROM THE     ***
      ***FIRST ROW ELEMENTS OF MATRIX M.  FOR EXAMPLE, BY PROPERTY (1),***
C     ***FORMULA (2) REDUCES TO                                          ***
C
C           M~ (+) = M              + M                         (4)
C            jk       1,k-(j-1)      1,n-(k+j)+2
C
C                   = M~ (+)            k=1, ..., n/2; j=1, ..., k
C                      kj
C
C     ***WHERE ADDITION OF THE COEFFICIENTS OF THE FIRST ROW OF MATRIX ***
C     ***M IS REPLACED BY SUBTRACTION WHEN COMPUTING M~(-).              ***
C
C     ***      IF z IS AN EIGENVECTOR OF M~(+) WITH CORRESPONDING EIGEN-***
C     ***VALUE d, AND J IS THE IDENTITY MATRIX OF ORDER n/2 WITH ITS    ***
C     ***COLUMNS WRITTEN IN REVERSE ORDER, THEN
C
C                      [ z ]         [ z ]
C                    M [     ]  =  d [     ].                   (5)
C                      [Jz ]         [Jz ]
C
C     ***SIMILARLY, IF z IS AN EIGENVECTOR OF M~(-) WITH CORRESPONDING ***
C     ***EIGENVALUE d, THEN
C
C                      [ z ]         [ z ]
C                    M [     ]  =  d [     ].                   (6)
C                      [-Jz]         [-Jz]
C
C     ***      COMPUTATIONAL RESULTS SHOW THAT THE EIGENPROBLEM FOR      ***
C     ***A MATRIX THAT SATISFIES (1) CAN BE SOLVED BY COMPUTER AT LEAST***
C     ***FOUR TIMES FASTER THIS WAY THAN BY SOLVING THE EIGENPROBLEM    ***
C     ***DIRECTLY FROM M FOR LARGE VALUES OF n.                          ***
C
```

```
C  ***     FURTHERMORE, SINCE THE EIGENVALUES AND EIGENVECTORS OF     ***
C  ***M~(+) ARE COMPUTED INDEPENDENTLY OF THOSE FOR M~(-), ONLY       ***
C  ***A MAXIMUM OF (3/4)n**2 + n STORAGE LOCATIONS ARE USED TO AC-    ***
C  ***COMODATE THESE MATRICES, THEIR EIGENVALUES AND EIGENVECTORS,    ***
C  ***INSTEAD OF 2n**2 + n WHEN THE EIGENVALUES AND EIGENVECTORS ARE***
C  ***EVALUATED DIRECTLY FROM THE ORIGINAL MATRIX M. IN PARTICULAR,  ***
C  ***ONLY (1/4)n**2 + n LOCATIONS ARE USED INSTEAD OF n**2 + n TO   ***
   ***ACCOMODATE THESE MATRICES AND THEIR EIGENVALUES, IF ONLY THE    ***
C  ***EIGENVALUES OF MATRIX M ARE REQUIRED.                           ***
C
C  ***    THE SUB-PROGRAM PACKAGE CONSISTS OF SUBROUTINES MATRED,     ***
C  ***MTLSYM, MTLFUL.  A SUBROUTINE THAT SOLVES THE EIGENPROBLEM      ***
C  ***OF A REAL SYMMETRIC MATRIX IS REQUIRED.  IN IMPLEMENTING, SUB-***
C  ***ROUTINE MATRED, WE USE THE IMSL LIBRARY ROUTINE EIGRS [3] TO   ***
C  ***SOLVE THE EIGENPROBLEMS OF M~(+) AND M~(-); HOWEVER, ANY OTHER***
C  ***ROUTINE THAT COMPUTES THE EIGENVALUES AND (OPTIONALLY) THE EI-***
C  ***GENVECTORS OF A REAL SYMMETRIC MATRIX MAY BE USED.              ***
C
C  ***    SINCE ROUTINES THAT SOLVE THE EIGENPROBLEM OF A REAL SYM-  ***
C  ***METRIC MATRIX REQUIRE THAT THE MATRIX BE STORED IN EITHER       ***
C  ***SYMMETRIC STORAGE MODE (LOWER TRIANGLE STORED AS A ONE DIMEN-  ***
C  ***SIONAL ARRAY IN ROW ORDER) OR FULL STORAGE MODE, TWO SUBROU-    ***
C  ***TINES ARE PROVIDED FOR THIS PURPOSE.  SUBROUTINE MTLSYM WILL    ***
C  ***COMPUTE M~(+) AND M~(-) IN SYMMETRIC STORAGE MODE, WHILE SUB-  ***
C  ***TINE MTLFUL WILL COMPUTE THEM IN FULL STORAGE MODE.             ***
C
C
C                            REFERENCES
C
C  ***1.  GOLDSTEIN MARVIN,  REDUCTION OF THE EIGENPROBLEM FOR HER-  ***
C  ***MITIAN PERSYMMETRIC MATRICES,  MATH. COMP., V.38, JAN. 1974    ***
C
C  ***2.  GOLDSTEIN MARVIN, "FURTHER DECOMPOSITION OF THE PSEUDOIN-  ***
C  ***VERSE AND EIGENSYSTEM OF A HERMITIAN PERSYMMETRIC MATRIX," ACM***
C  ***SIGNUM NEWSLETTER, V. 13, NO. 1, MARCH 1978.                   ***
C
C  ***3.  "EIGRS," IMSL LIBRARY FOR VAX-11/780, EIGTH EDITION, 1980.***
C  *******************************************************************
C
C  <><><><><><><><><><><><><><><><><><><><><><><><><><><><><><><><><>
C
C          ->INPUT VARIABLES FROM CALLING PROGRAM:
C
C MR1      DOUBLE PRECISION REAL VECTOR OF AT LEAST LENGTH N IN THE
C          CALLING PROGRAM CONTAINING FIRST ROW OF MATRIX M.
C N        INTEGER VARIABLE - ORDER OF MATRIX M
C IFULST   INTEGER VARIABLE SWITCH - SET TO ZERO IF M~(+) OR M~(-) IS
C          STORED IN SYMMETRIC STORAGE MODE BY EIGRS; SET TO ONE IF
C          FULL STORAGE MODE.
C MTILFL   DOUBLE PRECISION REAL, TWO DIMENSIONAL SCRATCH ARRAY OF ORDER
C          IFUL, WHICH HOLDS M~(+) OR M~(-) IN FULL STORAGE MODE.
C IFUL     INTEGER VARIABLE - SET TO ONE-HALF THE DIMENSION OF MR1
C          IN THE CALLING PROGRAM IF IFULST=1; SET TO 1 IF IFULST=0.
C MTILSY   DOUBLE PRECISION REAL SCRATCH VECTOR DIMENSIONED AT LEAST
C          ((N**2 + 2*N))/8 IN CALLING PROGRAM TO HOLD M~(+) OR
C          M~(-) IN SYMMETRIC STORAGE MODE IF IFULST=0; IF IFULST=1
C          THIS VECTOR SHOULD BE DIMENSIONED LENGTH ONE IN THE CAL-
C          LING PROGRAM.

C IVEC     INTEGER VARIABLE. IF IVEC IS ZERO, THEN ONLY EIGENVALUES
C          WILL BE COMPUTED; IF ONE, THEN EIGENVECTORS ALSO.
C IZ       INTEGER VARIABLE - ROW DIMENSION OF Z ARRAY IN CALLING PRO-
C          PROGRAM THAT HOLDS EIGENVECTORS OF M~(+) AND M~(-). IZ IS
C          ONE-HALF THE DIMENSION OF MR1 IN THE CALLING PROGRAM IF IVEC=
C          1; OTHERWISE, IZ IS ONE.
C WK       DOUBLE PRECISION WORKSPACE VECTOR OF AT LEAST LENGTH N.
C
C          ->OUTPUT TO CALLING PROGRAM:
C
C D        DOUBLE PRECISION REAL VECTOR OF AT LEAST LENGTH N
C          IN THE CALLING PROGRAM, WHICH HOLDS THE EIGENVALUES OF
C          MATRIX M.
C Z        DOUBLE PRECISION MATRIX OF IZ ROWS AND 2*IZ COLUMNS IN
C          THE CALLING PROGRAM.  CONTROL IS RETURNED TO THE INVOKING
C          PROGRAM WITH THE EIGENVECTORS OF M~(+) IN COLUMNS 1,...,N/2
C          OF MATRIX Z, AND THE EIGENVECTORS OF M~(-) IN COLUMNS
C          N/2+1,...,N, WITH CORRESPONDING EIGENVALUES IN VECTOR D.
C
C  <><><><><><><><><><><><><><><><><><><><><><><><><><><><><><><><><>
```

```
C
      INTEGER N,IER,IFULST,IFUL,IZ,IVEC,NDIV2,IPOINT
      DOUBLE PRECISION MR1,MTILFL,MTILSY,D,Z,WK,SIGN
      DIMENSION MR1(1),MTILFL(IFUL,1),MTILSY(1),D(1),Z(IZ,1),WK(1)
C
C                                   OMPUTE ORDER OF REDUCED MATRICES
C
      NDIV2=N/2
C
C                                   SET SIGN TO COMPUTE M~(+)
C
      SIGN=1.0 D0
C
C                                   FIRST TIME THRU LOOP SOLVE EIGEN-
C                                   PROBLEM FOR M~(+); SECOND TIME FOR
C                                   M~(-)
C
      DO 40 I=1,2
         IPOINT=(I-1)*NDIV2+1
         IF(IFULST.LT.1)GO TO 10
            GO TO 20
C        THEN COMPUTE USING SYMMETRIC STORAGE MODE
   10       CALL MTLSYM(NDIV2,MR1,MTILSY,SIGN)
            CALL EIGRS(MTILSY,NDIV2,IVEC,D(IPOINT),Z(1,IPOINT),IZ,WK,IER
     +      )
            GO TO 30
C        ELSE COMPUTE USING FULL STORAGE MODE
   20       IVEC=IVEC+10
C
C                                   SINCE IMSL ROUTINE EIGRS REQUIRES
C                                   THAT MTILFL BE EXACTLY N/2 BY N/2
C                                   THE FOURTH ARGUMENT IN THE MTLFL
C                                   CALL HAS BEEN REPLACED BY NDIV2.
C                                   RESTORE THIS ARGUMENT TO IFUL IF
C                                   REQUIRED BY AN EIGENVALUE ROUTINE.
C
            CALL MTLFUL(NDIV2,MR1,MTILFL,NDIV2,SIGN)
            CALL EIGRS(MTILFL,NDIV2,IVEC,D(IPOINT),Z(1,IPOINT),IZ,WK,IER
     +      )
   30    CONTINUE
C

C                                   SET SIGN TO COMPUTE M~(-)
C
         SIGN=-1.0 D0
C
   40 CONTINUE
C
      RETURN
      END
```

```
      SUBROUTINE MTLSYM(NDIV2,MR1,MTILDA,SIGN)
C     *************************************************************
C     ***      THIS SUBROUTINE RETURNS M~(+) OR M~(-) STORED IN   ***
C     ***SYMMETRIC STORAGE MODE IN THE ONE-DIMENSIONAL, DOUBLE    ***
C     ***PRECISION ARRAY MTILDA.  M~(+) AND M~(-) ARE COMPUTED    ***
C     ***FROM THE FIRST ROW OF MATRIX M, NAMELY MR1, WHEN SIGN    ***
C     ***IS +1 AND -1, RESPECTIVELY.                              ***
C     *************************************************************
      INTEGER I,J,K,NDIV2,N,L1,L2
      DOUBLE PRECISION MR1,MTILDA,SIGN
      DIMENSION MR1(1),MTILDA(1)
C
      N=NDIV2*2
C
      I=0
      DO 20 K=1,NDIV2
         DO 10 J=1,K
            I=I+1
            L1=K-J+1
            L2=N+2-K-J
            MTILDA(I)=MR1(L1)+MR1(L2)*SIGN
   10    CONTINUE
   20 CONTINUE
C
      RETURN
      END
```

```
      SUBROUTINE MTLFUL(NDIV2,MR1,MTILDA,IFUL,SIGN)
C     *************************************************************
C     ***      THIS SUBROUTINE RETURNS M~(+) OR M~(-) STORED IN   ***
C     ***FULL STORAGE MODE IN THE TWO-DIMENSIONAL, DOUBLE PRE-    ***
C     ***CISION ARRAY MTILDA.   M~(+) OR M~(-) ARE COMPUTED FROM***
C     ***THE FIRST ROW OF MATRIX M, NAMELY MR1, WHEN SIGN IS      ***
C     ***+1 AND -1, RESPECTIVELY.                                 ***
C     *************************************************************
C
      INTEGER N,NDIV2,I,IFUL,J,K,L1,L2
      DOUBLE PRECISION MR1,MTILDA,SIGN
      DIMENSION MR1(1),MTILDA(IFUL,1)
C
      N=NDIV2*2
C
      DO 20 K=1,NDIV2
         DO 10 J=1,K
            L1=K-J+1
            L2=N+2-K-J
            MTILDA(J,K)=MR1(L1)+MR1(L2)*SIGN
            MTILDA(K,J)=MTILDA(J,K)
   10    CONTINUE
   20 CONTINUE
C
      RETURN
      END
```

A SUBROUTINE PACKAGE FOR THE EFFICIENT SOLUTION OF
THE EIGENPROBLEM OF REAL SYMMETRIC TOEPLITZ MATRICES
Marvin J. Goldstein, Computer and Information Services Dept.
William Babson, Computer and Information Services Dept.
TM No. 831172
UNCLASSIFIED

DISTRIBUTION LIST

External:

Mr. I. L. Avrunin
David Taylor Naval Ship
Research & Development
Center
Headquarters
Bethesda, MD 20084

Dr. Martin Schultz
Research Center for Scientific Comp.
Yale Univ.
P. O. Box 2158 Yale Station
New Haven, CT 06520

Mr. Ken Kunec
Naval Data Automation
Command
Washington Navy Yard
Washington, DC 20374

Internal:

Code 00	CAPT Ailes
01	E.L. Messere
01A	John Keegan
039	Dr. D.M. Viccone
10	Dr. W.A. VonWinkle
101	Dr. E. Eby
02	D.C. Kindilien
0211	R.H. Bernier
02	NL Library (3 copies)
021311	NPT Library
20	W.L. Clearwaters
234	R.J. Conley
32	A. Lesick
	Dr. N.L. Owsley
	Dr. John Ianniello
	J. K. Sullivan
	H. Axtell
	R.C. Cox
	James Pearson
	L.C. Ng
	J.V. Sanchis
	B. Helme, Jr.
	Dr. C.H. Sherman
	W. Fox
	B.G. Buehler
	A.W. Ellenthorpe
	Dr. Peter G. Cable
	Dr. S. Ko
	D.T. Porter
	W.H. Wharton
	T. Anderson
	J. Ionata
	M. Kuuznitz
	J. Gannon
	C. Bowman
	P.R. Miner
	Rosemary Molino
	Jose Munoz
	J. Shores
	R. Leask
	Russell Christman
33	Dr. A.H. Nuttal
	J.B. Paniszczyn
	Dr. F.R. DiNapoli
	E.P. Jensen
	E.G. Kanabis
	S.C. Gerengher
	J. Gregor
	W.A. Goldman
	P.M. Anchors
	J.J. Higgs
	I.B. Cohen
	D.W. Counsellor
	Dr. R. Radliniski
	D.A. Fingerman
	R.J. Trembley
Code 33	J.E. Miller
	Dr. H.F. Dwyer
	Dr. C. Carter
	A. Quazi
	Peter Stahl
	J. Wolcin
	Dr. M.B. Moffett
	C.J. Becker
	F.C. Gannon
	R.J. MacDonald
	J. Saikowski
	G. Botseas
	Dr. D. Lee
	J. Nordquist
	E.R. Robinson
	H. Weinberg
	H. Sternberg
	Dr. David Wood
	Michael Fecher
34	J.R. Katan
	Dr. P.E. Fessenden
	C.N. Flemings
	K.F. Hafner
	John Casey
	Anthony Bruno
	D. Dixon
35	Tom Conrad
	L. Cabral
36	R.P. Eidimtas
	H.A. Rosen
	R.V. Cherry
	W.J. Ryan
	J. Griffin
	D.G. Blundell
	M.P. Lydon
	S.I. Wax
	S.E. Ashton
	Raymond McMahon
	S. Meyers
37	C.M. Curtis
	Q. Huynh
38	J.E. Sims
401	A.D. Carlson
	J.H. Clark
	Dr. A.J. Kalinowski
	Dr. R.G. Kasper
	R.R. Manstan
	E.L. McLaughlin
	R.S. Munn
	C.W. Nebelung
	Dr. J.S. Patel
	B.A. Radley
	A.Y. Shigematsu
	M.A. Tucchio
402	M. Berger
4111	Samuel Horvitz
4331	B.L. Antrim
	S.B. Walsh
5202	Stephan Schady
60	Dr. D. Bordelon
	Dr. Jeffrey Cohen
70	Cort R. DeVoe
701	R. Wilson Npt
	R. Streit
	D. McCue
	J. Babiec Npt
71	G. Elias Npt
	M. Goldstein (15 Copies)
	D. Drinkard
	D. Aker
	N. Sulinski
	W. Babson (15 Copies)
	E. Montavon
	J. Sikorski
	Louis Ledoux Npt
	R. Johnson
	H. Eley
	Cynthia Borges Npt
	Marcel Nadeau Npt
	John Ventura Npt
	Phoebe Liu Npt
	John Sabulis Npt
72	Gordon Daglieri Npt
	S. Schneller Npt
	J. Auwood
	T. Wheeler
	R. Warren Npt
	P. Breslin Npt
	A. Alfiero
	R. Clark
	J. Gribbin
73	D. Quigley
	S. Capizzano
	J. MacDonald
	R. Pingree
	R. Cote
	R. Forget
	A. Sullivan
	J. Surdo
74	M. Lee
	C. Brockway
	M. Becker
	L. Doyle
	J. Koonce
	J. Pappadia
	A. Levander
	Norman Dube Npt
	A. Blau Npt
	D. Stanhope Npt
	R. Hoy Npt
	N. Bradbury Npt